Prof. Dr. P. Kunkel
Fakultät für Mathematik und Informatik
Universität Leipzig

SS 2022

Numerische Optimierung

Implementation 3

Für NLSCON ergibt sich etwa als Problembeschreibung:

```
      IMPLICIT REAL*8(A-H,O-Z)
      PARAMETER (N=5,M=52,MFIT=M,RTOL=1.D-5)
      PARAMETER (LIWK=999,LRWK=9999)
      DIMENSION X(N),XSCAL(N),FI(MFIT),FSCAL(MFIT),IOPT(50)
      DIMENSION IWK(LIWK),RWK(LRWK)
      DATA X/1.D0,3.D0,0.25D0,0.2D0,0.25D0/
      DATA XSCAL/5*1.D0/
      DATA FI/3.1,3.2,3.3,3.5,3.6,3.8,3.8,3.9,3.9,4.0,3.9,4.0,4.0,
     &        4.0,4.2,4.0,3.9,4.0,3.8,3.7,3.4,3.5,3.4,3.3,3.0,3.1,
     &        2.9,3.0,2.7,2.6,2.6,2.4,2.3,2.2,2.0,2.1,2.1,1.8,2.1,
     &        2.1,2.0,2.1,2.3,2.3,2.3,2.4,2.5,2.6,3.0,2.9,3.0,2.9/
      DATA FSCAL/52*1.D0/
      DATA IOPT/50*0/
      EXTERNAL FCN,JAC
      IOPT(3)=1
      IOPT(11)=2
      IOPT(13)=2
      IOPT(15)=2
      CALL NLSCON(N,M,MFIT,FCN,JAC,X,XSCAL,FI,FSCAL,RTOL,IOPT,
     &   IERR,LIWK,IWK,LRWK,RWK)
      STOP
      END
C
      SUBROUTINE FCN(N,M,MCON,X,F,IFAIL)
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION X(N),F(M)
      FMOD(T)=(A*DCOS(2.D0*PI*(T-C))+B)*(1.D0+D*(1.D0-DEXP(-RLAM*T)))
      PI=4.D0*DATAN(1.D0)
```

```
      A=X(1)
      B=X(2)
      C=X(3)
      D=X(4)
      RLAM=X(5)
      DO 1 I=1,M
      T=I/52.D0
      F(I)=FMOD(T)
    1 CONTINUE
      RETURN
      END
C
      SUBROUTINE JAC(N,M,MCON,X,DFDX,IFAIL)
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION X(N),DFDX(M,N)
      DFA(T)=DCOS(2.D0*PI*(T-C))*(1.D0+D*(1.D0-DEXP(-RLAM*T)))
      DFB(T)=1.D0+D*(1.D0-DEXP(-RLAM*T))
      DFC(T)=2.D0*PI*A*DSIN(2.D0*PI*(T-C))*(1.D0+D*(1.D0-DEXP(-RLAM*T)))
      DFD(T)=(A*DCOS(2.D0*PI*(T-C))+B)*(1.D0-DEXP(-RLAM*T))
      DFLAM(T)=(A*DCOS(2.D0*PI*(T-C))+B)*D*T*DEXP(-RLAM*T)
      PI=4.D0*DATAN(1.D0)
      A=X(1)
      B=X(2)
      C=X(3)
      D=X(4)
      RLAM=X(5)
      DO 1 I=1,M
      T=I/52.D0
      DFDX(I,1)=DFA(T)
      DFDX(I,2)=DFB(T)
      DFDX(I,3)=DFC(T)
      DFDX(I,4)=DFD(T)
      DFDX(I,5)=DFLAM(T)
    1 CONTINUE
      RETURN
      END
```